

összeadás az Excelben - függvények

Mindennapi használatra, illetve mindennapi feladatok elvégzésére az Excel hat összeadó függvényt biztosít.

függvény neve	funkciója
SZUM	összeadás
ÖSSZESÍT	szelektív összeadás
RÉSZÖSSZEG	szelektív összeadás
SZUMHA	összeadás egy feltétellel
SZUMHATÖBB	összeadás több feltétellel
SZORZATÖSSZEG	szorzás, a szorzatok összeadása

A SZUM függvény az argumentumaival meghatározott számokat adja össze. Argumentumai lehetnek állandók, cella- és tartomány-hivatkozások és kifejezések. Maximális számuk kettőszáz-ötvenöt. A függvény csak a kifejezés-argumentumai eredményeként kapott IGAZ logikai értéket és szöveggént tárolt számokat veszi figyelembe, a hivatkozás-argumentumaival meghatározottakat nem.

	A	B	C	D	E
1	szám	számok	logikai	szöveg	
2	100	200	IGAZ	11	
3		300			
4		400			
5					
6					

Tehát a $SZUM(A2; B2:B4; C2:D2)$ képlet eredménye 1000 lesz, mert a C2-ben álló logikai értéket és a D2-ben szöveggént tárolt számot a SZUM függvény nem veszi figyelembe. De a $SZUM(A2; B2:B4; HA(2 * 2 = 4; IGAZ); SZÖVEG(2 / 2; "#"))$ képlet már 1002-t ad eredményül, mert a két kifejezés-argumentum eredménye is része az összegnek.

képlet	eredmény
$SZUM(A2; B2:B4; C2:D2)$	1000 (100 + 200 + 300 + 400)
$SZUM(A2; B2:B4; HA(2 * 2 = 4; IGAZ); SZÖVEG(2 / 2; "#"))$	1012 (100 + 200 + 300 + 400 + 1 + 11)

Ha sok tartományt kell argumentumként megadnunk, akkor a tartományok kijelölését ne szakítsuk meg a pontosvesszők beírásával, mert azt a program automatikusan megteszi! Tehát, [1] beírjuk az egyenlőségjelet és a SZUM-ot. Utóbbi a Tab billentyűvel visszük be a képletbe, mert akkor a nyitó zárójelet se nekünk kell begépelnünk, [2] kijelöljük az első tartományt, [3] lenyomjuk a Ctrl billentyűt és addig nyomva tartjuk, amíg ki nem jelöltük az utolsó tartományt is... Ugye azt mindenki tudja, hogy az egyetlen függvényt tartalmazó képlet, amit nevezhetnénk akár egytényezős képletnek is, mint például $=SZUM(< tartományok >)$, záró zárójelét sem kell nekünk begépelni? Azt a program automatikusan megteszi a szerkesztés lezárását követően.

A SZUM függvény hibaértéket ad eredményül, ha hivatkozás argumentumának valamely cellájában hibaérték áll vagy bármely kifejezés-argumentumának kiértékelése hibaértéket eredményez. Az eredmény azonos az első hibaértékkel.

	A	B	C	D	E
1	1	1	1	1	
2	#HIÁNYZIK	1	#NULLA!		1
3	1	#NÉV?	1	1	1
4	1	#HIV!	#SZÁM!		1
5	#ÉRTÉK!	1	1	#ZÉRÓOSZTÓ!	
6	1	1	1	1	1
7	#HIÁNYZIK	#NÉV?	#NULLA!	#ZÉRÓOSZTÓ!	
8					
9					

A hetedik sorban álló SZUM függvények hivatkozás-argumentumával meghatározott hibaértékek megakadályozzák az oszlop-összegek képzését: =SZUM(A1:A6), =SZUM(B1:B6), =SZUM(C1:C6)... Nézzünk két példát a kifejezés-argumentumok hibaértékeire is.

képlet	eredmény
SZUM(ABC(7); ABS("koko"))	#NÉV? (#NÉV? ; #ÉRTÉK!)
SZUM(A3:A4 ; C1 ; HA(2 > 1 ; "koko"))	#ÉRTÉK! (2 ; 1 ; koko)

Az első képlet nem igényel magyarázatot, de a második már igen. A képletben a kifejezés-argumentum kiértékelése nem hibaértéket, hanem egy karakterláncot eredményez. Tehát, a SZUM függvény hibaértékes eredményét nem csak egyik argumentumának hibája, hanem a számmá nem konvertálható szöveg is okozhatja.

Szűrt- és rejtett sorokat tartalmazó táblázatban a függvény működik ugyan, de nem úgy, ahogy várnánk: a nem látható számokat is szerepelteti az összegben.

	A	B	C
1	napok	számok	
2	hétfő	1	
3	kedd	1	
4	szerda	1	
5	csütörtök	1	
6	péntek	1	
7	szombat	1	
8	vasárnap	1	
9		7	
10		=SZUM(B2:B8)	
11			

	A	B	C
1	napok	számok	
3	kedd	1	
4	szerda	1	
6	péntek	1	
9		7	
10		=SZUM(B2:B8)	
11			
12			
13			
14			
15			

	A	B	C
1	napok	számok	
2	hétfő	1	
4	szerda	1	
6	péntek	1	
9		7	
10		=SZUM(B2:B8)	
11			
12			

A képen látható táblázatot napokra szűrtük, de a B9 cellában álló SZUM függvény eredményét ez nem befolyásolta. A helyzet, akkor sem változik, ha a szűrést a sorok rejtésével helyettesítjük.

A SZUM függvényt részösszegek táblázatban sem tudjuk használni, mert a függvény nem ismeri fel a részösszegeket tartalmazó SZUM-os cellákat és ezért a részösszegeket is szerepelteti a végösszegben.


	A	B	C	D
1		számok		
2	január	1		
3	február	1		
4	március	1		
5	I. negyedév	3	=SZUM(B2:B4)	
6	április	1		
7	május	1		
8	június	1		
9	II. negyedév	3	=SZUM(B6:B8)	
10	I. félév	12		
11			=SZUM(B2:B9)	
12				
13				

A SZUM függvény mentségére legyen mondva, a legtöbb statisztikai függvény hasonló helyzetekben hasonlóan viselkedik.

A problémát az ÖSSZESÍT és a RÉSZÖSSZEG függvényekkel orvosolhatjuk. Kezdjük a szűrt- illetve a rejtett sorokat tartalmazó táblázatokkal. A függvények első argumentuma egy szám: a helyettesítendő függvény azonosítója. A SZUM azonosítója az ÖSSZESÍT-ben a 9-es, a RÉSZÖSSZEG-ben a 109-es. Utóbbi esetében, azért nem a „sima” 9-es, mert az a rejtett sorokban álló számokat is hozzáadja az összeghez. A szűréssel elrejtett sorokban állókat viszont alpból kihagyja.



Az ÖSSZESÍT viszont nem tesz különbséget a szűréssel vagy a rejtéssel láthatatlanná tett számok között. Ha a második argumentuma 5-ös, akkor a nem látható számok nem szerepelnek az összegben.



A függvények további argumentumai az összeg-tartományok hivatkozásai. Természetesen, most csak a függvények összeadó funkciójával foglalkozunk.

Ahogy a képen is látható az ÖSSZESÍT a hibaértékes tartományokban is tud összeadni, ha erre a második argumentumában álló 6-ossal utasítjuk. Sőt a 7-essel nem csak a hibaértékeket, de a nem látható számokat is mellőzi.

Részösszeges tartomány SZUM-os részösszegeit sem az ÖSSZESÍT sem a RÉSZÖSSZEG függvény-nel nem tudjuk kizárni az összegből. Egyet tehetünk lecseréljük a SZUM-okat RÉSZÖSSZEG-re.

	A	B	C	D
1	hónapok	számok		
2	január	1		
3	február	1		
4	március	1		
5	I. negyedév	3	=SZUM(B2:B4)	
6	április	1		
7	május	1		
8	június	1		
9	II. negyedév	3	=SZUM(B6:B8)	
10	I. félév			
11				
12				

? X

Keresés és csere

Keresés Cseré

Keresett szöveg: SZUM(Nincs formátum beállítva Formátum...

Csere erre: RÉSZÖSSZEG(109 ; Nincs formátum beállítva Formátum...

Hatáskör: Munkalap Kis- és nagybetű különbözik

Sorrend: Soronként Ha ez a teljes cella tartalma

Keresés helye: Képletek

Az összes cseréje Cseré

	A	B	C	D	E
1	hónapok	számok			
2	január	1			
3	február	1			
4	március	1			
5	I. negyedév	3	=RÉSZÖSSZEG(109 ; B2:B4)		
6	április	1			
7	május	1			
8	június	1			
9	II. negyedév	3	=RÉSZÖSSZEG(109 ; B6:B8)		
10	I. félév	6			
11			=RÉSZÖSSZEG(109 ; B2:B9)		
12					
13					

A fenti részösszeg-végösszeg felépítést az ÖSSZESÍT függvénnyel is kialakíthatjuk, sőt a függvények „felismerik” egymást, a részösszegek képletében. Természetesen az ÖSSZESÍT csak akkor, ha erre a második argumentumával utasítjuk.

Ne felejtsek el azt a lehetőséget sem, hogy az *automatikus statisztikai képlet* szolgáltatással, népi nevén az *autoszum*-mal, **össze tudjuk adni a SZUM-os részösszegeket**.

A SZUM függvény nem válogat a számok között, az argumentumaival meghatározott összes számot összeadja. Ezzel szemben, a bemutatásra kerülő további függvények már lehetőséget kínálnak az összeadandó számok szelektálására is.

Mielőtt folytatnánk, tisztázzunk néhány fogalmat. Az összeadásra kerülő számok általában egy táblázatban állnak. Mondjuk egy oszlopban. Ha a táblázat első sorában és első oszlopában címkék állnak,

számlaszám	ügyfél	város	számla	egyenleg
05-79-41-33	Abonyi Olimpia	Székesfehérvár	Ultimate	-163 000
00-00-04-34	Ács Rozália	Pécs	Gold	84 000
80-27-46-88	Adorján Mihály	Székesfehérvár	Gold	-308 000
05-59-69-48	Adorján Szabrina	Pécs	Elit	-205 000
01-73-05-96	Agócs Norbert	Székesfehérvár	Standard	59 000
00-00-04-47	Ambrus Báborka	Budapest	Elit	-451 000
00-00-06-83	Angyal Katalin	Székesfehérvár	Gold	69 000
00-00-02-61	Asolti Hermina	Győr	Standard	-352 000
00-72-48-79	Bacsó Csenge	Pécs	Gold	289 000

Például a képen megjelölt szám a 05-59-69-48 számú számla egyenlege. Tehát a szám neve „egyenleg”, azonosítója a „számlaszám”. Az egyenleg szempontjából, a sorában álló további adatok, az egyenleg tulajdonságai.

A SZUMHA függvény a feltételes összeadás eszköze. Az összeadandó számok körét egy feltétellel szabályozzuk, amely vonatkozhat magára a számra vagy egyik tulajdonságára. A függvény három argumentuma, sorrendben a következő. [1] Feltétel-tartomány hivatkozása: azoknak az adatoknak a cellái, amelyekben a feltétel teljesülését vizsgálni kell. [2] Feltétel: relációból és a viszonyítási alapból álló logikai kifejezés. [3] Összeg-tartomány hivatkozása: az összeadandó számok cellái. Ha a feltétel közvetlenül a számokra irányul, akkor a számok tartományát nem kell még egyszer megadnunk.

Nézzünk egy-egy példát a függvény alkalmazására. Kíváncsiak vagyunk a pozitív egyenlegű számlák egyenlegeinek összegére. Tehát a feltétel most közvetlenül az összeadandó számokra irányul: SZUMHA(<egyenlegek cellái> ; <nagyobb mint nulla>). De, ha azt kérdezzük, hogy mennyi a pécsi számlák egyenlegeinek összege, akkor már a feltétel az összeadandó számok egy tulajdonságára vonatkozik: SZUMHA(<városok cellái> ; <pécs> ; <egyenlegek cellái>).

A függvény második argumentumának írásmodja kötött és szabályai számosak. Megpróbálom értelmesen ismertetni őket. A feltétel két részből áll, reláció és viszonyítási alap. Kivéve, ha a reláció egyenlőség. Azt nem jelöljük. Tehát a reláció nélküli feltétel-argumentummal azonosságot írunk elő! A relációs jelek sorrendje kötött. Nem egyenlő: <>. Kisebbségi vagy egyenlő: <=. Nagyobb egyenlő: >=.

	A	B	C	D	E	F	G
1	név	város	éves	szám			
2	Örs	Vác	22	1			
3	Ica	Ózd	21	1			
4	Éva	Érd	18	1			
5	Ibi	Tab	20	1			
6	Pál	Göd	19	1			
7	Ili	Mór	20	1			
8	Ede	Érd	20	1			
9	Tas	Vác	18	1			
10	Ági	Vác	22	1			
11							
12							
13							
14							
15							

=SZUMHA(B2:B10 ; "<>vác" ; D2:D10) » 6
 =SZUMHA(B2:B10 ; "vác" ; D2:D10) » 3
 =SZUMHA(C2:C10 ; "<=19" ; D2:D10) » 3
 =SZUMHA(C2:C10 ; ">=20" ; D2:D10) » 6

A viszonyítási alap lehet szám, dátum, idő, szöveg, logikai érték vagyis állandó, cella-hivatkozás és kifejezés.

	A	B	C	D	E	F	G
1	név	város	éves	szám			
2	Örs	Vác	22	1			
3	Ica	Ózd	21	1			
4	Éva	Érd	18	1			
5	Ibi	Tab	20	1			
6	Pál	Göd	19	1			
7	Ili	Mór	20	1			
8	Ede	Érd	20	1			
9	Tas	Vác	18	1			
10	Ági	Vác	22	1			
11							
12							
13							
14							

=SZUMHA(C2:C10 ; ">20" ; D2:D10) » 3
 =SZUMHA(C2:C10 ; ">" & C5 ; D2:D10) » 3
 =SZUMHA(C2:C10 ; ">" & ÁTLAG(C2:C10) ; D2:D10) » 3

A feltétel idézőjelek között áll. Kivéve, ha a reláció egyenlő és a viszonyítási alap szám vagy logikai érték vagy hivatkozás vagy kifejezés.

	A	B	C	D	E	F	G
1	név	éves	családos	szám			
2	Örs	22	IGAZ	1			
3	Ica	21	HAMIS	1			
4	Éva	18	IGAZ	1			
5	Ibi	20	IGAZ	1			
6	Pál	19	HAMIS	1			
7	Ili	20	IGAZ	1			
8	Ede	20	HAMIS	1			
9	Tas	18	HAMIS	1			
10	Ági	22	HAMIS	1			
11							
12							
13							
14							

=SZUMHA(B2:B10 ; ">20" ; D2:D10) » 3
 =SZUMHA(B2:B10 ; 20 ; D2:D10) » 3
 =SZUMHA(C2:C10 ; HAMIS ; D2:D10) » 5
 =SZUMHA(B2:B10 ; B5 ; D2:D10) » 3
 =SZUMHA(B2:B10 ; ÁTLAG(B2:B10) ; D2:D10) » 3

Szám-állandó tartalmazhat tizedesvesszőt, százalék jelet és megadhatjuk a számot normál alakban is. További formátumoktól, például Ft vagy HUF tartózkodjunk! Egyenlő reláció esetén a feltétel csak akkor teljesül, ha a viszonyítási alap és a cella tartalma a pontosság szerint azonos. Magyarán, ha a viszonyítási alap 1,2, akkor sem a 1,20001, sem a 1,199 nem teljesíti a feltételt.

	A	B	C	D	E	F	G
1	név	város	állapot	szám			
2	Örs	Vác	34,999%	1			
3	Ica	Ózd	42,4%	1			
4	Éva	Érd	30,20%	1			
5	Ibi	Tab	35%	1			
6	Pál	Göd	39,2%	1			
7	Ili	Mór	38%	1			
8	Ede	Érd	35,01%	1			
9	Tas	Vác	28%	1			
10	Ági	Vác	35,001%	1			
11							
12							
13							
14							

=SZUMHA(C2:C10 ; ">0,35" ; D2:D10) » 5
 =SZUMHA(C2:C10 ; ">35%" ; D2:D10) » 5
 =SZUMHA(C2:C10 ; ">3,5e-1" ; D2:D10) » 5
 =SZUMHA(C2:C10 ; 0,35 ; D2:D10) » 1

A dátum-állandóban a dátumegységeket elválaszthatjuk ponttal, kötőjellel vagy perjellel. Idő-állandóban csak a kettőspontot ismeri fel a függvény. Az elválasztó karakter után szóköz is állhat. A dátumban az évszázad két számjegye, valamint a hónap-szám és a nap-szám vezető nullája elhagyható. Idő-állandóban az órák vezető nullja hagyható el. A dátum-feltétel viszonyítási alapjában pozitív egész számot, az idő-feltételben egynél kisebb, pozitív számot is megadhatunk. Magyarul, használhatjuk a dátum és időpont illetve időegység szám-értékét is. Az idő-feltételben megadott ó:pp formátumú idő automatikusan nulla másodperccel egészül ki: ó:pp:00. Az azonosság ellenőrzésénél ezt figyelembe kell vennünk!

	A	B	C	D	E	F	G
1	név	született	érkezett	szám			
2	Örs	1998.11.06.	18:31:11	1			
3	Ica	1999.05.04.	11:42:40	1			
4	Éva	2002.02.15.	3:52:03	1			
5	Ibi	2000.07.27.	3:49:55	1			
6	Pál	2001.05.15.	19:46:05	1			
7	Ili	2000.12.22.	18:33:16	1			
8	Ede	2000.09.12.	4:42:34	1			
9	Tas	2002.06.04.	10:40:18	1			
10	Ági	1998.01.01.	9:16:11	1			
11							
12							
13							
14							

=SZUMHA(B2:B10 ; ">2001.01.01" ; D2:D10) » 3
 =SZUMHA(B2:B10 ; ">01-01-01" ; D2:D10) » 3
 =SZUMHA(B2:B10 ; ">1/ 1" ; D2:D10) » 3
 =SZUMHA(C2:C10 ; ">12:01" ; D2:D10) » 3
 =SZUMHA(C2:C10 ; ">0,5" ; D2:D10) » 3
 =SZUMHA(C2:C10 ; "<=9:16" ; D2:D10) » 3

Vegyes adattípusú tartományban a szöveg adattípusú adatokat a csillag karakterrel választhatjuk ki: "*" . A feltétel tagadása ("<>*") az összes nem szöveg adattípusú adat elfogadását jelenti, beleértve az üres és a hibaértéket tartalmazó cellákat is.

	A	B	C	D	E	F	G
1	név	város	vegyes	szám			
2	Örs	Vác	szöveg	1			
3	Ica	Ózd	21	1			
4	Éva	Érd	IGAZ	1			
5	Ibi	Tab	20	1			
6	Pál	Göd	szöveg	1			
7	Ili	Mór	szöveg	1			
8	Ede	Érd	20	1			
9	Tas	Vác		1			
10	Ági	Vác	#NÉV?	1			
11							
12							

=SZUMHA(C2:C10 ; "*" ; D2:D10) » 3
 =SZUMHA(C2:C10 ; "<>*" ; D2:D10) » 6

A függvény a karakterláncok vizsgálatánál a kis- és nagybetűket nem különbözteti meg. A feltételben használhatunk helyettesítő karaktereket is. Csillag (*): nem meghatározott számú karakter. Kérdőjel (?): egy karakter. A helyettesítő funkciót ellátó karaktereket a hullámvonal (~) karakter segítségével kereshetjük. A csillagot: ~*. A kérdőjelet: ~?.

	A	B	C	D	E	F	G
1	név	város	nyelv	szám			
2	Örs	Pomáz	angol	1			
3	Ica	Gyula		1			
4	Éva	Tokaj		1			
5	Ibi	Dabas	angol	1			
6	Pál	Hévíz	angol?	1			
7	Ili	Lenti	olasz?	1			
8	Ede	Tokaj	francia	1			
9	Tas	Pomáz		1			
10	Ági	Pomáz	német?	1			
11							
12							

=SZUMHA(B2:B10 ; "pomáz" ; D2: D10) » 3
 =SZUMHA(B2:B10 ; "?o???" ; D2:D10) » 5
 =SZUMHA(C2:C10 ; "*~?" ; D2:D10) » 3

Ha a viszonyítási alap hivatkozás vagy kifejezés, akkor a reláció idézőjelek között áll és end (&) jellel kapcsolódik a viszonyítási alaphoz.

	A	B	C	D	E	F	G
1	név	város	éves	szám			
2	Örs	Vác	22	1			
3	Ica	Ózd	21	1			
4	Éva	Érd	18	1			
5	Ibi	Tab	20	1			
6	Pál	Göd	19	1			
7	Ili	Mór	20	1			
8	Ede	Érd	20	1			
9	Tas	Vác	18	1			
10	Ági	Vác	22	1			
11							
12							
13							
14							

=SZUMHA(B2:B10 ; "<>" & B2 ; D2: D10) » 6
 =SZUMHA(C2:C10 ; "<=" & C5 ; D2:D10) » 6
 =SZUMHA(C2:C10 ; ">" & ÁTLAG(C2:C10) ; D2:D10) » 3

Az üres cellát a feltételben nulla hosszúságú szöveggel, a tartalommal rendelkezőt pedig az idézőjelek között álló, viszonyítási alap nélküli, nem egyenlő relációval adjuk meg.

	A	B	C	D	E	F	G
1	név	város	angol	szám			
2	Örs	Vác	felső	1			
3	Ica	Ózd		1			
4	Éva	Érd	alap	1			
5	Ibi	Tab	alap	1			
6	Pál	Göd		1			
7	Ili	Mór		1			
8	Ede	Érd	felső	1			
9	Tas	Vác	közép	1			
10	Ági	Vác	közép	1			
11							
12							

=SZUMHA(C2:C10 ; "" ; D2: D10) » 3
 =SZUMHA(C2:C10 ; "<>" ; D2:D10) » 6

Tehát a függvény három argumentuma sorrendben: a feltétel-tartomány hivatkozása, a feltétel és az összeg-tartomány hivatkozása. A függvény szépen sorban, megvizsgálja a feltétel-tartomány celláit, teljesíti-e a cellában álló adat a feltételt. Ha igen, akkor megkeresi az összeg-tartomány azonos pozíciójában álló celláját és a benne álló számot hozzáadja az összeghez. Ha a feltétel közvetlenül a számra irányul, akkor még keresgélnie sem kell.

	A	B	C	D	E	F	G
1		létszám		termelés			
2		január	február	január	február		
3	üzem 1	10	9	30	40		
4	üzem 2	7	11	50	50		
5	üzem 3	8	12	50	30		
6	üzem 4	12	11	40	30		
7							
8			140				
9		=SZUMHA(B3:C6 ; "<10" ; D3:E6)					
10							
11							
12							

Elárulok egy titkot. A függvény az összeg-tartomány deklarációjának csak az első celláját veszi figyelembe. Ettől a cellától kezdve „méri ki” a feltétel-tartomány sor- és oszlopszámával azonos összeg-tartományt. Tehát, ha csak ezt írjuk: =SZUMHA(B3:C6 ; "<10" ; D3), a függvény működni fog. De ez, azt is jelenti, hogyha az összeg-tartományt rosszul adjuk meg, akkor a függvény nem küld hibaüzenetet, nem ad hibaértéket eredményül, csak éppen a kapott összeg nem megfelelő. És figyelem, figyelem! A feltétel szintaktikai hibája esetén is ugyanez történik!

A SZUMHATÖBB függvény a SZUMHA továbbfejlesztett és javított változata. Nemcsak egy, de akár száz-húszon-hét feltételt is megszabhatunk közvetlenül az összeadandó számok- vagy valamely tulajdonságuk szelektálására. Első argumentuma az összeg-tartomány hivatkozása. További argumentumai párban állnak: feltétel-tartomány hivatkozása, feltétel. A feltétel-tartományok sor- és oszlopszámának azonosnak kell lenniük az összeg-tartomány sor- és oszlopszámával. Az előző bekezdésben ismertetett „laza működésnek” ennél a függvénynél már nyoma sincs: az eltérő alakú tartományok hibaértéket a szintaktikai hibák hibaüzenetet generálnak. A feltétel argumentumok írásmódja azonos a SZUMHA feltételének írásmódjával.

A függvény egy munkamenetben az azonos pozícióban álló feltétel-tartományok celláit vizsgálja: tartalmuk teljesíti-e a tartományukhoz rendelt feltételt. Ha az összes feltétel teljesült, akkor az összeg-tartomány azonos pozíciójú cellájának száma hozzáadódik az összeghez.

	A	B	C	D	E	F	G
1	név	város	típus	szám			
2	Örs	Vác	vevő	1			
3	Ica	Ózd	eladó	2			
4	Éva	Érd	vevő	1			
5	Ibi	Vác	eladó	3			
6	Pál	Vác	eladó	1			
7	Ili	Mór	vevő	1			
8	Ede	Érd	felső	1			
9	Tas	Vác	eladó	2			
10	Ági	Vác	vevő	1			
11							
12				5			
13	=SZUMHATÖBB(D2:D10 ; B2:B10 ; "vác" ; C2:C10 ; "eladó" ; D2:D10 ; ">1")						
14							
15							

A fenti táblázatban a váci eladásokat összesítettük, de csak azokat, amelyeknek a mennyisége meghaladja az egyet.

A szorzatok összeadása mindennapi művelet. Gondoljunk csak egy vásárlásra: több tétel eltérő darabszámmal és eltérő egységárral. Mennyit kell majd fizetnünk?

	A	B	C	D	E	F
1	megnevezés	mennyiség	egységár	ár		
2	fekete szilva	0,385 kg	799 Ft/kg	307,615	=B2*C2	
3	jumbo tojás	10 db	37,9 Ft/db	379	=B3*C3	
4	Korfú kocka	0,023 kg	7 999 Ft/kg	183,977	=B4*C4	
5	Schweppes	1,5 l	249 Ft/l	373,5	=B5*C5	
6						
7			1 245 HUF			
8	=TÖBBSZ.KEREKÍT(SZUM(D2:D5) ; 5)					
9						
10						

Az összeget a TÖBBSZ.KEREKÍT függvénnyel az öt többszörösére kerekítettem. A függvény első argumentumával a kerekítendő számot a másodikkal a kerekítés alapját deklaráljuk.

[népművelés] Egy képletet aposztróffal kezdve tudunk szöveggé megjeleníteni a cellában: '<képlet>'. De ezt teszi a KÉPLETSZÖVEG függvény is, amely az egyetlen argumentumaként megadott cella képletét alakítja át szöveggé. [népművelés vége]

Szóval van két tartományunk „mennyiség” és „egységár”. A tartományok azonos pozíciójú számait összesoroztuk, majd a szorzatokat összeadtuk. Pontosan ezt teszi a SZORZATÖSSZEG függvény is. Argumentumai a tartományok hivatkozásai. Nézzünk még egy példát, amin megfigyelhetjük a függvény használatát.

	A	B	C	D	E	F	G	H	I	J	
1		FO-850	MO-663	KI-223	TE-982	FO-954	ME-260	DE-107	GO-296		
2	alkalmazás naponta (alkalom)	1	5	3	1	2	3	1	3		
3	alkalmanként (darab)	2	1	1	1	2	2	2	1		
4	kúra hossza (nap)	14	3	7	1	5	14	7	3		
5	egy tabletta ára (forint)	530	600	730	780	1120	490	700	1120		
6											
7	gyógyszerköltség	123 390	=SZORZATÖSSZEG(B2:I2 ; B3:I3 ; B4:I4 ; B5:I5)								
8											
9											

Szándékosan választottam olyan példát, amelyben az összeszorozandó tartományok nem függőleges tájolásúak. Ez az elrendezés ritka, ezért gyakran zavarba hozza a gyanútlan Excel-használót...



margitfalvi.arpad@gmail.com